

Operacións con expresións

Maple dispón de moitas ferramentas para manipular expresións matemáticas. Algunhas das funcións máis utilizadas son:

Función **expand**:

intenta simplificar unha expresión en forma de sumas de produtos de funcións máis sinxelas.

expand(expr, expr1, expr2, ..., exprn)

onde **expr** é a expresión matemática que se quere simplificar e **expr1, expr2, ..., exprn** son argumentos opcionais que indican as subexpresións que non queremos que se simplifiquen. Algúns exemplos:

```
> expand((x+1)*(x+2));
```

$$x^2 + 3x + 2 \quad (1.1)$$

```
> expand(((x+1)*(x+3)*x)/(x+2)); # só simplifica o numerador da expresión
```

$$\frac{x^3}{x+2} + \frac{4x^2}{x+2} + \frac{3x}{x+2} \quad (1.2)$$

```
> expand(sin(x+y));
```

$$\sin(x) \cos(y) + \cos(x) \sin(y) \quad (1.3)$$

```
> expand(cos(2*x));
```

$$2 \cos(x)^2 - 1 \quad (1.4)$$

```
> expand(exp(a+ln(b)));
```

$$e^a b \quad (1.5)$$

```
> expand((x+1)*(y+z));
```

$$xy + xz + y + z \quad (1.6)$$

```
> expand(ln(x/(1-x)^2)); # só se expanden as expresións cando é posíbel
```

$$\ln\left(\frac{x}{(1-x)^2}\right) \quad (1.7)$$

```
> expand((x+1)*(y+z), x+1); # o segundo argumento (x+1) especifica a subexpresión que desexamos que non se simplifique.
```

$$(x+1)y + (x+1)z \quad (1.8)$$

Función **combine**: realiza a tarefa inversa a **expand**. Combina varias expresións para conseguir unha expresión máis compacta ou reducida.

combine(f) ou **combine(f, n)**

onde **f** é unha expresión, lista ou set de expresións a combinar, **n** é o nome, lista ou set de nomes que indican o tipo de elementos que contén a expresión para que MAPLE sepa que tipo de regras ten que utilizar. Valores usuais son: **trig, exp, ln, power, etc**. As regras de combinación en cada caso son:

Para **trig**

$$\sin(a) \sin(b) ==> 1/2 \cos(a-b) - 1/2 \cos(a+b)$$

$$\begin{aligned}\sin(a)\cos(b) & \implies 1/2\sin(a-b) + 1/2\sin(a+b) \\ \cos(a)\cos(b) & \implies 1/2\cos(a-b) + 1/2\cos(a+b)\end{aligned}$$

Para **exp**

$$\exp(x)\exp(y) \implies \exp(x+y)$$

$$\exp(x)^y \implies \exp(x^y)$$

$$\exp(x+n\ln(y)) \implies y^n\exp(x) \text{ where } n \text{ is an integer}$$

Para **ln**

$$a\ln(x) \implies \ln(x^a) \text{ (provided } a \cdot \text{argument}(x) = \text{argument}(x^a) \text{)}$$

$$\ln(x)+\ln(y) \implies \ln(x*y) \text{ (provided } \text{argument}(x*y) = \text{argument}(x) + \text{argument}(y) \text{)}$$

> **combine(sin(x)*cos(y)+cos(x)*sin(y), trig); #especificamos que use as regras trigonométricas**

$$\sin(x+y) \quad (1.9)$$

> **exp(x+3*ln(y));**

$$e^{x+3\ln(y)} \quad (1.10)$$

> **combine(%, exp);**

$$y^3 e^x \quad (1.11)$$

> **exp(sin(a)*cos(b))*exp(cos(a)*sin(b));**

$$e^{\sin(a)\cos(b)} e^{\cos(a)\sin(b)} \quad (1.12)$$

> **combine(%, [trig, exp]); #especificamos que utilice un lista de regras de combinación**

$$e^{\sin(a+b)} \quad (1.13)$$

Función simplify: é o comando xeral de simplificación en MAPLE. Ó igual que **combine** podemos especifica-lo tipo de simplificación que desexamos (trig, exp, ln,...). Se non se especifica nada MAPLE intenta aplica-lo maior número de regras de simplificación posíbel. Nalgúns casos devolve o mesmo resultado que **expand**.

> **expresion:=sin(x)^2+ln(2*x)+cos(x)^2;**

$$\text{expresion} := \sin(x)^2 + \ln(2x) + \cos(x)^2 \quad (1.14)$$

> **simplify(expresion);**

$$1 + \ln(2) + \ln(x) \quad (1.15)$$

> **simplify(expresion, trig); # aqui so se aplicas as regras de simplificación trigonométricas. Obsérvese a diferenza co exemplo anterior.**

$$1 + \ln(2x) \quad (1.16)$$

Función normal: resulta útil para aplicar a expresións alxebraicas que conteñan **fraccións**. A función normal convirte a fracción na forma normal factorizada, é dicir, fracción da forma numerador/denominador onde numerador e denominador son polinomios primos (indivisibeis) con coeficientes enteiros.

> **restart;**

> **expresion:=(x^2-y^2)/(x-y)^3;**

$$\text{expresion} := \frac{x^2 - y^2}{(x - y)^3} \quad (1.17)$$

> **normal(expresion);**

$$\frac{x+y}{(x-y)^2} \quad (1.18)$$

```
> normal((x^2-1)/(x-1));
```

$$x+1 \quad (1.19)$$

```
> normal(sin(x*(x+1)-x));
```

$$\sin(x^2) \quad (1.20)$$

Se desexamos que normal expanda no seu resultado tanto o numerador como denominador hai que proporcionarlle como segundo argumento *expanded*:

```
> normal(expressao, expanded);
```

$$\frac{x+y}{x^2-2xy+y^2} \quad (1.21)$$

Función factor: permite descompoñer un polinomio en factores (para descompoñer un número utilízase o comando **ifactor**).

```
> factor(6*x^2+18*x-24);
```

$$6(x+4)(x-1) \quad (1.22)$$

```
> factor(6);
```

$$6 \quad (1.23)$$

```
> ifactor(6);
```

$$(2)(3) \quad (1.24)$$

```
> expr1:=1/(x^2-1)+1/(x^2+3*x+2);
```

$$\text{expr1} := \frac{1}{x^2-1} + \frac{1}{x^2+3x+2} \quad (1.25)$$

```
> factor(expr1);
```

$$\frac{2x+1}{(x+2)(x+1)(x-1)} \quad (1.26)$$

```
> ifactor(902/24);
```

$$\frac{(11)(41)}{(2)^2(3)} \quad (1.27)$$

Función convert: descompón unha fracción alxebrica en fraccións simples

convert(expr, form, arg3, ...)

onde **expr** é a fracción a descompoñer, **form** indica o tipo de descomposición, **arg3** indica a variábel respecto da cal se realiza a descomposición (opcional se non hai máis dunha variábel). A lista de valores que pode toma-lo argumento **form** pódese consultar na axuda.

```
> restart;
```

```
> f := (x^3+x)/(x^2-1);
```

$$f := \frac{x^3+x}{x^2-1} \quad (1.28)$$

```
> convert(f, parfrac, x);
```

$$x + \frac{1}{x-1} + \frac{1}{x+1} \quad (1.29)$$